線形代数 I (2013): 逆行列の計算

照井 章（筑波大学 数理物質系）
2013 年 5 月 31 日

線形代数 I（化学類対象）

基本行列の定義

基本行列の定義を以下の通り行う。

(1) 行列の第 j 行の c 倍: multiplyRow[n, j, c]

multiplyRow[n, j, c] --- n 次単位行列の 第 j 行を c 倍する

```
multiplyRow[n_, j_, c_] := ReplacePart[IdentityMatrix[n], {j, j} → c]
```

(2) 行列の第 i 行と第 j 行の交換: exchangeRows[n, i, j]

exchangeRows[n, i, j] --- n 次単位行列の第 i 行と第 j 行を入れ換える

```
exchangeRows[n_, i_, j_] :=
 ReplacePart[IdentityMatrix[n], {{i, i} → 0, {i, j} → 1, {j, i} → 1, {j, j} → 0}]
```

(3) 行列の第 i 行に第 j 行の c 倍を加える: rowAdd[n, i, j, c]

rowAdd[n, i, j, c] --- n 次単位行列の第 i 行に第 j 行の c 倍を加える

```
rowAdd[n_, i_, j_, c_] := ReplacePart[IdentityMatrix[n], {i, j} → c]
```

正則行列の逆行列の計算（教科書 p. 41, 例題 2.3）

$$A = \begin{pmatrix} 1 & 2 & 1 \\ 2 & 4 & 1 \\ 1 & 1 & 2 \end{pmatrix}$$

の逆行列を求める。

```
A = {{1, 2, 1}, {2, 4, 1}, {1, 1, 2}}
```

```
{{1, 2, 1}, {2, 4, 1}, {1, 1, 2}}
```

```
MatrixForm[A]
```

$$\begin{pmatrix} 1 & 2 & 1 \\ 2 & 4 & 1 \\ 1 & 1 & 2 \end{pmatrix}$$

最初に行列 $(A \mid E_3)$ を定義する。これを AA とおく。

```
AA = Table[Join[A[[i]], IdentityMatrix[3][[i]]], {i, 3}]
```

```
{{1, 2, 1, 1, 0, 0}, {2, 4, 1, 0, 1, 0}, {1, 1, 2, 0, 0, 1}}
```

```
MatrixForm[AA]
```

$$\begin{pmatrix} 1 & 2 & 1 & 1 & 0 & 0 \\ 2 & 4 & 1 & 0 & 1 & 0 \\ 1 & 1 & 2 & 0 & 0 & 1 \end{pmatrix}$$

まず第1列の行消去を行う。
第2行に第1行の –2 倍を加える基本行列を Q1 とおく。

```
Q1 = rowAdd[3, 2, 1, -2]
```

```
{{1, 0, 0}, {-2, 1, 0}, {0, 0, 1}}
```

```
MatrixForm[Q1]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ -2 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Q1 を AA の左側からかける。計算結果を A1 とおく。

```
A1 = Q1.AA
```

```
{{1, 2, 1, 1, 0, 0}, {0, 0, -1, -2, 1, 0}, {1, 1, 2, 0, 0, 1}}
```

```
MatrixForm[A1]
```

$$\begin{pmatrix} 1 & 2 & 1 & 1 & 0 & 0 \\ 0 & 0 & -1 & -2 & 1 & 0 \\ 1 & 1 & 2 & 0 & 0 & 1 \end{pmatrix}$$

第3行に第1行の –1 倍を加える基本行列を Q2 とおく。

```
Q2 = rowAdd[3, 3, 1, -1]
```

```
{{1, 0, 0}, {0, 1, 0}, {-1, 0, 1}}
```

```
MatrixForm[Q2]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ -1 & 0 & 1 \end{pmatrix}$$

Q2 を A1 の左側からかける。計算結果を A2 とおく。

```
A2 = Q2.A1
```

```
{{1, 2, 1, 1, 0, 0}, {0, 0, -1, -2, 1, 0}, {0, -1, 1, -1, 0, 1}}
```

```
MatrixForm[A2]
```

$$\begin{pmatrix} 1 & 2 & 1 & 1 & 0 & 0 \\ 0 & 0 & -1 & -2 & 1 & 0 \\ 0 & -1 & 1 & -1 & 0 & 1 \end{pmatrix}$$

第1列の消去は終了。
今度は第2列の消去を行うが、(2, 2) 成分が 0 なので、先に第2行と第3行を交換する。
第2行と第3行を交換する基本行列を Q3 とおく。

```
Q3 = exchangeRows[3, 2, 3]
```

```
{{1, 0, 0}, {0, 0, 1}, {0, 1, 0}}
```

```
MatrixForm[Q3]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 0 & 1 \\ 0 & 1 & 0 \end{pmatrix}$$

Q3 を A2 の左側からかける。計算結果を A3 とおく。

```
A3 = Q3.A2
```

```
{{1, 2, 1, 1, 0, 0}, {0, -1, 1, -1, 0, 1}, {0, 0, -1, -2, 1, 0}}
```

```
MatrixForm[A3]
```

$$\begin{pmatrix} 1 & 2 & 1 & 1 & 0 & 0 \\ 0 & -1 & 1 & -1 & 0 & 1 \\ 0 & 0 & -1 & -2 & 1 & 0 \end{pmatrix}$$

第2行を –1 倍する基本行列を Q4 とおく。

```
Q4 = multiplyRow[3, 2, -1]
```

```
{{1, 0, 0}, {0, -1, 0}, {0, 0, 1}}
```

```
MatrixForm[Q4]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Q4 を A3 の左側からかける。計算結果を A4 とおく。

```
A4 = Q4.A3
```

```
{{1, 2, 1, 1, 0, 0}, {0, 1, -1, 1, 0, -1}, {0, 0, -1, -2, 1, 0}}
```

```
MatrixForm[A4]
```

$$\begin{pmatrix} 1 & 2 & 1 & 1 & 0 & 0 \\ 0 & 1 & -1 & 1 & 0 & -1 \\ 0 & 0 & -1 & -2 & 1 & 0 \end{pmatrix}$$

第1行に第2行の −2 倍を加える基本行列を Q5 とおく。

```
Q5 = rowAdd[3, 1, 2, -2]
```

{{1, -2, 0}, {0, 1, 0}, {0, 0, 1}}

```
MatrixForm[Q5]
```

$$\begin{pmatrix} 1 & -2 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Q5 を A4 の左側からかける。計算結果を A5 とおく。

```
A5 = Q5.A4
```

{{1, 0, 3, -1, 0, 2}, {0, 1, -1, 1, 0, -1}, {0, 0, -1, -2, 1, 0}}

```
MatrixForm[A5]
```

$$\begin{pmatrix} 1 & 0 & 3 & -1 & 0 & 2 \\ 0 & 1 & -1 & 1 & 0 & -1 \\ 0 & 0 & -1 & -2 & 1 & 0 \end{pmatrix}$$

第2列の消去は終了。今度は第3列の消去を行う。
第3行を −1 倍する基本行列を Q6 とおく。

```
Q6 = multiplyRow[3, 3, -1]
```

{{1, 0, 0}, {0, 1, 0}, {0, 0, -1}}

```
MatrixForm[Q6]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -1 \end{pmatrix}$$

Q6 を A5 の左側からかける。計算結果を A6 とおく。

```
A6 = Q6.A5
```

{{1, 0, 3, -1, 0, 2}, {0, 1, -1, 1, 0, -1}, {0, 0, 1, 2, -1, 0}}

```
MatrixForm[A6]
```

$$\begin{pmatrix} 1 & 0 & 3 & -1 & 0 & 2 \\ 0 & 1 & -1 & 1 & 0 & -1 \\ 0 & 0 & 1 & 2 & -1 & 0 \end{pmatrix}$$

第2行に第3行の 1 倍を加える基本行列を Q7 とおく。

```
Q7 = rowAdd[3, 2, 3, 1]
```

{{1, 0, 0}, {0, 1, 1}, {0, 0, 1}}

```
MatrixForm[Q7]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 1 \\ 0 & 0 & 1 \end{pmatrix}$$

Q7 を A6 の左側からかける。計算結果を A7 とおく。

```
A7 = Q7.A6
```

{{1, 0, 3, -1, 0, 2}, {0, 1, 0, 3, -1, -1}, {0, 0, 1, 2, -1, 0}}

```
MatrixForm[A7]
```

$$\begin{pmatrix} 1 & 0 & 3 & -1 & 0 & 2 \\ 0 & 1 & 0 & 3 & -1 & -1 \\ 0 & 0 & 1 & 2 & -1 & 0 \end{pmatrix}$$

第1行に第3行の –3 倍を加える基本行列を Q8 とおく。

```
Q8 = rowAdd[3, 1, 3, -3]
```

{{1, 0, -3}, {0, 1, 0}, {0, 0, 1}}

```
MatrixForm[Q8]
```

$$\begin{pmatrix} 1 & 0 & -3 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Q8 を A7 の左側からかける。計算結果を A8 とおく。

```
A8 = Q8.A7
```

{{1, 0, 0, -7, 3, 2}, {0, 1, 0, 3, -1, -1}, {0, 0, 1, 2, -1, 0}}

```
MatrixForm[A8]
```

$$\begin{pmatrix} 1 & 0 & 0 & -7 & 3 & 2 \\ 0 & 1 & 0 & 3 & -1 & -1 \\ 0 & 0 & 1 & 2 & -1 & 0 \end{pmatrix}$$

これで A8 の左半分は E_3 に等しくなった。このとき、右半分が A^{-1} である。
実際にこのことを確かめる。
A8 の右半分のブロックをあらためて B とおく。

```
B = A8[[All, 4 ;; 6]]
```

{{-7, 3, 2}, {3, -1, -1}, {2, -1, 0}}

```
MatrixForm[B]
```

$$\begin{pmatrix} -7 & 3 & 2 \\ 3 & -1 & -1 \\ 2 & -1 & 0 \end{pmatrix}$$

$A\,B$ を計算すると、単位行列 E_3 に等しくなる。

```
A.B // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

$B\,A$ も、単位行列 E_3 に等しくなる。

```
B.A // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

以上により、$B = A^{-1}$が確かめられた。